

# Incremental Proof Development in Dafny with Module-Based Induction

Son Ho  
Inria  
France

Clément Pit-Claudel  
EPFL  
Switzerland

## Abstract

Highly automated theorem provers like Dafny allow users to prove simple properties with little effort, making it easy to quickly sketch proofs. The drawback is that such provers leave users with little control about the proof search, meaning that the small changes inherent to the iterative process of writing a proof often lead to unpredictable variations in verification time, and eventually hard-to-diagnose proof failures. This sometimes turns the boon of high automation into a curse, as instead of breaking early and showing unsolved goals to the user like in Coq, proofs tend to gradually become unstable until their verification time explodes. At this point, the absence of a proof context to investigate often leaves the user to a painful debugging session. In this paper, we show how to use Dafny modules to encode Coq-like induction principles to dramatically improve the stability and maintainability of proofs about inductive data structures.

## ACM Reference Format:

Son Ho and Clément Pit-Claudel. 2024. Incremental Proof Development in Dafny with Module-Based Induction. In *Proceedings of ACM Conference (Conference'17)*. ACM, New York, NY, USA, 6 pages. <https://doi.org/10.1145/nnnnnnn.nnnnnnn>

## 1 Introduction

Writing a mechanized proof generally implies working in an incremental manner, by first laying out a simplified version of the problem under study and shaping the proofs, then gradually complexifying the definitions while updating those proofs. For instance, to verify a compiler one might start with a simplified version of the source AST that omits the complex cases, before adding those cases one by one. Highly automated theorem provers like Dafny make the first iterations pleasant as the prover manages to discharge most of the then-simple proof obligations. Unfortunately, as the

problem gets more complicated, automated proofs can become unstable or simply break. Worse, most automated theorem provers only offer limited tools to debug unstable or failing proofs: unlike an ITP like Coq, an ATP like Dafny does not show a goal or a proof context along with each failure. In an attempt to recover the benefits of ITPs, we demonstrate in this short paper how to structure inductive proofs in Dafny by using modules to encode Coq-like induction principles. We report on our experience of iterating through the proofs of a prototype of a self-hosted compiler for the Dafny language, and on the benefits in terms of proof maintainability and stability.

## 2 Inductive Proofs in Dafny and Coq

*Comparing Dafny and Coq.* Let us start with a minimal example: a proof that list concatenation is associative, in Dafny and Coq (listing 1). In Dafny, we define the concatenation as `App` (“append”) and do the proof by induction in `Assoc`. Dafny automatically discharges the `Nil` case. In the `Cons` case, we use a recursive call (line 17) to invoke the induction hypothesis.

In Coq, we use `induction ls0 as ...` to invoke the induction principle that Coq automatically derived from the definition of `ll`, which, after a call to `simpl1` to simplify the context, gives us two goals:

$$\Gamma : \text{Type}, ls_1 : ll \Gamma, ls_2 : ll \Gamma \vdash \text{app } ls_1 \ ls_2 = \text{app } ls_1 \ ls_2$$

for the `Nil` case, and:

$$\dots, IH : \forall ls_1 \ ls_2. \text{app } (\text{app } t \ ls_1) \ ls_2 = \text{app } t \ (\text{app } ls_1 \ ls_2)$$

$$\vdash \text{Cons } h \ (\text{app } (\text{app } t \ ls_1) \ ls_2) = \text{Cons } h \ (\text{app } t \ (\text{app } ls_1 \ ls_2))$$

for the `Cons` case.

From there, it is easy to determine how to invoke the induction hypothesis (`rewrite IH`, with unification filling in the arguments) in the `Cons` case. We finally conclude both goals by reflexivity. In contrast, in Dafny: 1. we have to write the inductive structure by hand; 2. we are not shown goals; 3. we cannot use unification to instantiate the induction hypothesis, and must instead specify arguments to the recursive call. This isn’t an issue when doing simple proofs, but is a significant burden when working on more realistic cases. In particular, Dafny doesn’t provide much information to the user when a proof breaks, while Coq displays the precise goal on which it got stuck. As a result, the user often spends a significant amount of time debugging broken proofs to understand *which* proof obligation failed, before actually spending time on fixing it.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from [permissions@acm.org](mailto:permissions@acm.org).

*Conference'17, July 2017, Washington, DC, USA*

© 2024 Association for Computing Machinery.

ACM ISBN 978-x-xxxx-xxxx-x/YY/MM... \$15.00

<https://doi.org/10.1145/nnnnnnn.nnnnnnn>

```

1 // Dafny
2 datatype LL<T> =
3 | Nil | Cons(h: T, t: LL<T>)
4
5 function App<T>(
6   l0: LL<T>, l1: LL<T>) : LL<T>
7 { match l0
8   case Nil => l1
9   case Cons(h, t) => Cons(h, App(t, l1)) }
10
11 lemma Assoc<T>(l0: LL<T>,
12   l1: LL<T>, l2: LL<T>)
13 ) ensures App(App(l0, l1), l2)
14   == App(l0, App(l1, l2))
15 { match l0
16   case Nil => // Nothing to do
17   case Cons(h, t) => Assoc(t, l1, l2); }
18
19 (* Coq *)
20 Inductive ll T :=
21 | Nil | Cons : T -> ll T -> ll T.
22
23 Fixpoint app {T} (l0 l1: ll T) : ll T :=
24   match l0 with
25   | Nil => l1
26   | Cons h t => Cons h (app t l1)
27   end.
28
29 Lemma assoc {T} (l0 l1 l2 : ll T) :
30   app (app l0 l1) l2 = app l0 (app l1 l2).
31 Proof.
32   induction ls0 as [|hd t1 IH]; intros; simpl.
33   + reflexivity. (* Nil case *)
34   + rewrite IH. reflexivity. (*Cons case *)
35 Qed.

```

**Listing 1.** Proof that list concatenation is associative (top: Dafny, bottom: Coq).

Another issue is proof duplication. We notice that in practice most proofs about lists follow the same structure: we perform an induction, and in the `Cons` case call the induction hypothesis on the tail of the list; writing the inductive structure and specifying the arguments to the recursive call by hand leads to a lot of boilerplate in Dafny.

One final issue is proof evolution. In Coq, adding an additional constructor, maybe `Snoc : ll T → T → ll T`, would lead to a failed proof with a new unsolved goal clearly identifying what is missing. Dafny, in contrast, would first try to derive a contradiction in the missing case (`Snoc`), and having failed to do that would report the missing case (without showing a proof context). This contradiction proof can be costly: in extreme cases, e.g. with many other constructors, it can fail to complete and we may simply get an unspecific proof failure for the whole lemma, without further details. A solution might be to forbid the users from omitting cases in a match; this is however not desirable in practice, as programmers often use this convenience to avoid considering many irrelevant cases in their proofs.

```

abstract module ListInduction {
  predicate P<T>(ls: LL<T>)

  lemma Induct_Nil<T>()
    ensures P<T>(Nil)

  lemma Induct_Cons<T>(
    h: T, t: LL<T>
  ) requires P(t)
    ensures P(Cons(h, t))

  lemma Induct<T>(ls: LL<T>)
    ensures P(ls) {
  match ls
  case Nil => Induct_Nil<T>();
  case Cons(h, t) => Induct(t); Induct_Cons(h, t); } }

module AppAssoc refines ListInduction {
  predicate P ...
  { forall l1, l2 ::
    App(App(ls, l1), l2) ==
    App(ls, App(l1, l2)) }

  // "... is Dafny syntax to reuse the
  // signatures defined in the module
  // ListInduction
  lemma Induct_Nil ... {}
  lemma Induct_Cons ... {} }

```

**Listing 2.** Proof of associativity of list concatenation with a module-based induction principle.

*Using an induction principle in Dafny.* We propose to structure the proofs with an induction principle, by which we factor out the structure of the proofs and decompose the various proof obligations, leading to less mundane work, a better debugging experience and finer control over verification times. The use of induction principles is inspired by provers like Coq which generate them for free; in the case of Dafny we have to write them by hand.<sup>1</sup>

We proceed by defining an induction principle for lists in the form of an abstract module (`ListInduction` in listing 2). We declare the target property that we wish to prove by induction through an abstract predicate `P`, together with the rules it must satisfy: `Induct_Nil` states that `P` must be true for the empty list, and `Induct_Cons` states that it must be true on non-empty lists provided it is true on their tails. Those abstract declarations act like holes: the corresponding proofs are to be filled later. Given those assumptions, we can prove by induction once and for all that `P` always holds (lemma `Induct`).

This abstract module provides a generic structure for all the inductive proofs for lists; in particular we can use it to prove associativity. We do so by defining a module named `AppAssoc` which refines `ListInduction`. This time, we have to

<sup>1</sup>The induction principle we introduce here is actually simplistic for the purpose of clarity. For more realistic versions, see Section 3.

fill in the blanks: we state the associativity property by providing a definition for  $P$ , and write proofs for `Induct_Nil` and `Induct_Cons`; as Dafny manages to discharge the proofs automatically, they are empty. Note that because the proofs are empty, Dafny actually allows us to omit those lemmas, which is in practice very useful when there are a lot of trivial cases; Dafny would however report an error if it fails to prove on its own a lemma that we omitted. The theorem we want is finally given by `AppAssoc.Induct`, that we can use without additional work. Using an induction principle for this simple example might seem overkill; we illustrate the benefits on more realistic examples in the next sections.

### 3 Applying the Induction Principle on Mini-Dafny

#### 3.1 Verifying `IsPure`

We explained how to define and use an induction principle on the simplistic example of list concatenation. Let us now illustrate how it can be adapted to a more interesting example, namely verifying micro-passes of a compiler for a simple language based on Dafny that we call mini-Dafny. We make the whole development available in the companion artifact [2]. We adapted this language from a work-in-progress verified compiler for the Dafny programming language [1]. The problem of verifying multiple compilation passes, which involved repeatedly proving inductive theorems involving the same, big function (`InterpStmt` is around 1000 LoCs in [1]) provided the initial motivation for the present work.

```

1 function InterpStmt(s: Stmt, ctx: Context):
2   Result<(int, Context)> {
3   match s {
4   case Bind(bvar, bval, body) =>
5     // ':' below is a monadic bind
6     var (bvalv, ctx1) :- InterpStmt(bval, ctx);
7     var ctx2 := ctx1[bvar := bvalv];
8     var (bodyv, ctx3) :- InterpStmt(body, ctx2);
9     var ctx4 := ctx1 + (ctx3 - {bvar});
10    Success((bodyv, ctx4))
11
12  case Seq(s1, s2) =>
13    var (_, ctx1) :- InterpStmt(s1, ctx);
14    InterpStmt(s2, ctx1)
15
16  ... /* Omitted */ } }
```

We define the semantics of mini-Dafny with an interpreter (`InterpStmt`). For simplicity, values are integers, and we omit all side effects but in-place updates to local variables. `InterpStmt` takes as inputs a statement and a context, which is a map from variable names to integer values, and returns the result of evaluating the statement together with an updated context. In order to evaluate a variable declaration (`Bind` case), we first evaluate the bound value `bval` (line 6), where `:-` is a bind for the error monad, meaning the statement `var x :- y; st` is desugared to `match y { case Fail e => Fail e; case Return(x) => st; }`. We then augment the context with a new

binding for `bvar` (line 7), evaluate the body in this new context (line 8), and finally reset the value bound to `bvar` (if this binding exists in the initial context), so that the bound variable doesn't escape its scope (line 9). Specifically, at line 9, `ctx3 - {bvar}` is the map `ctx3` where we remove the binding for `bvar` (if it exists), and `ctx1 + (ctx3 - {bvar})` is `ctx1` extended with the bindings from `ctx3 - {bvar}` (if a binding exists in both maps, we take the one from `ctx3 - {bvar}`). Finally, evaluating a sequence of statements (`Seq` case) simply requires chaining contexts between the statements of the sequence.

Given those semantics for mini-Dafny, we can verify a first micro-pass which rewrites statements of the form  $\theta * s$  or  $s * \theta$  to  $\theta$ , provided  $s$  is pure (i.e., it doesn't update any local variable); note that in mini-Dafny we mix statements and expressions. We first define a predicate `IsPure(s: Stmt, locals: set<string>)` which states that statement  $s$  doesn't update variables but the ones listed in `locals`; we use `locals` to track variables bound in declarations and whose updates won't escape their scope. In particular, if `IsPure(s, {})` is true then  $s$  doesn't have side effects. Looking at the definition, a declaration (`Bind`) is pure if the bound statement `bval` only updates variables from `locals`, and if its body only updates variables from the set `{bvar} + locals`. An in-place update (`Assign`) is pure if it updates the value of a variable from `locals`. A sequence is pure if it is made of pure statements. We omit the other, straightforward cases. For instance, `x := 3` is not pure, while `var x := 0; x := 3` is pure because  $x$  is locally bound and won't escape its scope.

```

predicate IsPure(
  s: Stmt, locals: set<string> := {}) {
  match s
  case Bind(bvar: Var, bval: Stmt, body: Stmt) =>
    IsPure(bval, locals) && IsPure(body, {bvar} + locals)
  case Assign(avar, aval) =>
    avar in locals && IsPure(aval, locals)
  case Seq(s1, s2) =>
    IsPure(s1, locals) && IsPure(s2, locals)
  ... /* Omitted */ }
```

Now suppose we want to prove the correctness of `IsPure`, meaning that if a statement is pure in the sense of `IsPure` then evaluating it leaves the context unchanged. As the proof proceeds by induction over mini-Dafny statements<sup>2</sup> we introduce an induction principle to reason over the mini-Dafny AST.

```

predicate P(st: S, s: Stmt)
predicate P_Step(st: S, s: Stmt, st1: S, v: V)
... // Some definitions omitted

lemma P_Step_Sound(st: S, s: Stmt, st1: S, v: V)
  requires P_Step(st, s, st1, v)
  ensures P(st, s)

lemma InductSeq_Step(
```

<sup>2</sup>mini-Dafny doesn't have loops, which would require induction over semantic derivations.

```

st: S, s: Stmt, s1: Stmt, s2: Stmt, st1: S, v1: V)
requires s == Seq(s1, s2)
requires P_Step(st, s1, st1, v1)
requires P(st1, s2)
ensures P(st, s)

... // Omitted: lemmas for the various inductive cases

```

Defining an induction principle for mini-Dafny requires a bit more work than for lists. We require an abstract  $P(\text{st}: S, s: \text{Stmt})$  predicate which states the target property for statement  $s$  in state  $\text{st}$ . Importantly, we use an abstract type  $S$  for the states, because the user might want to carry more information than just a single context. We do the same for values, for similar reasons. We also require an auxiliary predicate  $P\_Step$  to mention intermediary steps of execution. The  $P\_Step(\text{ctx}: S, s: \text{Stmt}, \text{ctx1}: S, v: \text{int})$  predicate states that evaluating  $s$  starting in state  $\text{ctx}$  succeeds and yields a new state  $\text{ctx1}$  and a value  $v$ ; as we need to link it to  $P$  somehow, we also require that  $P\_Step$  implies  $P$  through the (abstract) lemma  $P\_Step\_Sound$ . We then decompose the inductive cases into precise lemmas. For instance,  $InductSeq\_Step$  states that, if the target property holds for the execution of  $s1$  starting in  $\text{st}$ , and also holds for the execution of  $s2$  starting in the state resulting from executing  $s1$ , then it holds for the whole sequence  $s1; s2$ , starting in  $\text{st}$ . We finally show how to use this induction principle for the proof of correctness of  $IsPure$ .

```

1 datatype S =
2   S(locals: set<string>, ctx: Context)
3 type V = int
4
5 predicate SameCtxs(locals, ctx, ctx1) {
6   && ctx1.Keys == ctx.Keys
7   && ctx1 - locals == ctx - locals }
8
9 predicate P_Step(st, s, st1, v) {
10  && IsPure(s, st.locals)
11  && st.ctx.Keys >= st.locals
12  && InterpStmt(s, st.ctx) == Success((v, st1.ctx))
13  && st1.locals == st.locals
14  && SameCtxs(st.locals, st.ctx, st1.ctx) }
15
16 predicate P(st, s) {
17  IsPure(s, st.locals) ==>
18  st.ctx.Keys >= st.locals ==>
19  match InterpStmt(s, st.ctx) {
20    case Failure _ => true
21    case Success(_, ctx1) =>
22      SameCtxs(st.locals, st.ctx, ctx1)
23  } }

```

We define the state as a pair of a set of variable names and a context. The predicate  $P$  states that if  $s$  only updates variables from  $\text{st.local}s$  according to  $IsPure$  (line 17), and if the context has bindings for the variables listed in  $\text{st.local}s$  (line 18), then evaluating  $s$  starting in  $\text{st.ctx}$  yields (if it succeeds) a context which is unchanged but on the variables listed in  $\text{st.local}s$  (line 22). We need the condition  $\text{st.ctx.Keys}$

$>= \text{st.local}s$  to ensure that  $InterpStmt$  won't fail while accessing a variable listed in  $\text{st.local}s$  because it is undefined. The predicate  $P\_Step$  is similar to  $P$  except for the condition that executing  $s$  in  $\text{st}$  must succeed, yielding  $v$   $\text{st1}$  (line 12), and for the conjunctions which replace implications (this is slightly technical: suffices to say that  $P\_Step$  must *unconditionally* state that the execution succeeds; we omit the rule which enforces this). Overall, instantiating the induction principle for  $IsPure$  is straightforward, and all the proofs go through automatically.

### 3.2 Experience Using the Induction Principle

We now report on our experience of using the induction principle in practice. We applied the induction principle to several proofs, namely: 1.  $IsPure$ ; 2.  $EliminateMulZero$ : a micro-pass which simplifies statements of the form  $0 * s$  or  $s * 0$  to  $0$ , provided  $s$  is pure; 3.  $UnchangedVar$ : a predicate which states that a specific variable is left unchanged by a statement.

We made several changes to the mini-Dafny AST in order to evaluate the cost of updating the proofs: 1. we updated  $Seq$  to contain an arbitrary number of statements ( $Seq(\text{Stmt}, \text{Stmt}) \rightarrow Seq(seq<\text{Stmt}>)$ ). 2. we updated  $Bind$  (and  $Assign$ ) to allow multiple declarations (assignments, respectively) at once.

We initially introduced  $UnchangedVar$  to reason about variable inlining when working on the more mature version of the compiler [1]. Though simple in appearance, this property is actually subtle and led to expensive proofs; we thus resorted to using a module-based induction principle. In practice, this approach allowed us to dramatically decrease the time we spent on maintaining the proofs.

*Factoring out proofs.* By using an induction principle we don't have to write the inductive structure of the proofs by hand, and even get automatic variable introduction; after we paid the cost of writing this principle, we thus recover similar advantages to using a tactic like  $induction$  in Coq. Instantiating the induction principle by providing definitions for the abstract declarations (e.g.,  $P$ ) requires some boilerplate, especially as we introduced more abstract definitions with each language extension. In practice, however, this work was straightforward, especially as mistakes done when instantiating the induction principle were easy to debug and fix, and in particular easier to fix than the version of the proofs which did not use an induction principle. We also noticed that our instantiations shared similarities: we might leverage this fact to reduce the work even further in the future.

*Easy debugging.* Because we wrote the rules required by the induction principle so that they are small and precise, we were able to quickly pinpoint the reasons behind a failure whenever a proof broke. In particular, Dafny would tell us which specific lemma (and thus inductive case) failed. This allowed us to easily fix proofs when updating the language, and proved useful when sketching the proofs in the first place, as we could quickly iterate by adjusting the way

we stated the properties we targeted to prove until we got them right.

*Smooth iterations.* Updating the mini-Dafny language required us to update the induction principle several times, either by modifying specific rule statements, or by adding more rules and abstract definitions. As a result, we could focus on specific changes while updating the proofs, and the fact that the rules are small and simple made them more stable. In practice, updating the mini-Dafny language only required us to provide definitions for the new declarations we introduced in the induction principle, and to add *one* assertion at one location in the proof of `VarUnchanged`, to guide the SMT solver in its search.

## 4 Related Work

Previous work explored the design of proofs robust to changes, typically by introducing abstractions and interfaces [7, 15, 27, 28]. Some work explored the problem of automating inductive proofs altogether by means of heuristics [10, 12, 14, 16, 20, 26]; in our case, we target properties which are usually too complex to be fully automated. Other work explored the problem of writing usable inversion theorems but for ITPs like Coq, for instance to control the size of the generated proof terms [21, 22]. An obvious way of overcoming the limitations described in Section 2 is to extend ATPs with tactics, as done in [9, 18, 19]. In the context of the present work extending the Dafny prover was however not an option, but it would be interesting to investigate how our approach compares with using tactics in a tool which supports both (like F\* or Why3). Interestingly, if the use of heavy automation has been promoted a lot to stabilize proofs when using ITPs [4, 6, 13, 23, 29], much less work went into the problem of stabilizing proofs which already relied on a high level of automation. We can however mention attempts to stabilize the proof search itself [17] or preserve VC transformations and proof attempts [5]. Related to the problem of proof maintenance, some recent work explored the problem of proof repair [24, 25], but in the context of tactic-based proof assistants like Coq and not in highly automated theorem provers like Dafny. Finally, it is worth noting that other works tackled the problem of studying language semantics in an ATP [3, 8, 11]. We however note that none of those targeted the verification of a multi-pass compiler for a realistic language, which on our side provided the original motivation for introducing our encodings of induction principles [1].

## 5 Conclusion

We demonstrated how to encode an induction principle in Dafny by means of an abstract module. By applying this induction principle to case studies taken from iterations over the proofs of correctness of a compiler for the mini-Dafny language, we showed that our technique has clear benefits

to help the user factor out and maintain proofs. In effect, by structuring inductive proofs we relieve the user from mundane work spent on structuring those proofs, and allow them to focus instead of their core. By decomposing inductive proof obligations into small and precise lemmas, we also make it easy to pinpoint the reason behind proof failures. As future work, we are planning to investigate how to automate the process of generating induction principles, as is done in ITPs like Coq.

## References

- [1] Dafny compiler bootstrap: a work-in-progress reimplementa-tion of dafny's compilers, in dafny. <https://github.com/dafny-lang/compiler-bootstrap>, 2023.
- [2] Incremental proof development in dafny with module-based induction - artifact. <https://doi.org/10.5281/zenodo.10553207>, 2024.
- [3] Benedikt Becker and Claude Marché. Ghost Code in Action: Automated Verification of a Symbolic Interpreter. In Supratik Chakraborty and Jorge A.Navias, editors, *VSTTE 2019 - 11th Working Conference on Verified Software: Tools, Techniques and Experiments*, volume 12031 of *Lecture Notes in Computer Science*, New York, United States, July 2019. URL: <https://inria.hal.science/hal-02276257>, doi:10.1007/978-3-030-41600-3\\_8.
- [4] Jasmin Blanchette, David Greenaway, Cezary Kaliszzyk, Daniel Kühlwein, and Josef Urban. A learning-based fact selector for isabelle/hol. *Journal of Automated Reasoning*, 57, 10 2016. doi:10.1007/s10817-016-9362-8.
- [5] François Bobot, Jean-Christophe Filliâtre, Claude Marché, Guillaume Melquiond, and Andrei Paskevich. Preserving user proofs across specification changes. In Ernie Cohen and Andrey Rybalchenko, editors, *Verified Software: Theories, Tools, Experiments*, pages 191–201, Berlin, Heidelberg, 2014. Springer Berlin Heidelberg.
- [6] Adam Chlipala. *Certified Programming with Dependent Types: A Pragmatic Introduction to the Coq Proof Assistant*. The MIT Press, 2013.
- [7] Jacek Chrząszcz. Implementing modules in the coq system. In David Basin and Burkhart Wolff, editors, *Theorem Proving in Higher Order Logics*, pages 270–286, Berlin, Heidelberg, 2003. Springer Berlin Heidelberg.
- [8] Martin Clochard, Jean-Christophe Filliâtre, Claude Marché, and Andrei Paskevich. Formalizing Semantics with an Automatic Program Verifier. In Dimitra Giannakopoulou and Daniel Kroening, editors, *6th Working Conference on Verified Software: Theories, Tools and Experiments (VSTTE)*, volume 8471 of *Lecture Notes in Computer Science*, Vienna, Austria, July 2014. Springer. URL: <https://inria.hal.science/hal-01067197>.
- [9] Sylvain Dailier, Claude Marché, and Yannick Moy. Lightweight interactive proving inside an automatic program verifier. *Electronic Proceedings in Theoretical Computer Science*, 284:1–15, nov 2018. URL: <https://doi.org/10.4204/eptcs.284.1>, doi:10.4204/eptcs.284.1.
- [10] Maulik A. Dave. Review of rippling: Meta-level guidance for mathematical reasoning cambridge tracks in theoretical computer science 56 by alan bundy, david basin, dieter hutter, and andrew ireland. *SIGACT News*, 42(4):21–23, dec 2011. doi:10.1145/2078162.2078166.
- [11] Nicolas Jeannerod, Claude Marché, and Ralf Treinen. A Formally Verified Interpreter for a Shell-like Programming Language. In *VSTTE 2017 - 9th Working Conference on Verified Software: Theories, Tools, and Experiments*, volume 10712 of *Lecture Notes in Computer Science*, Heidelberg, Germany, July 2017. URL: <https://hal.science/hal-01534747>.
- [12] Moa Johansson, Lucas Dixon, and Alan Bundy. Case-analysis for rippling and inductive proof. volume 6172, pages 291–306, 07 2010. doi:10.1007/978-3-642-14052-5\_21.

- [13] Cezary Kaliszyk and Josef Urban. Learning-assisted automated reasoning with flyspeck (vol 53, pg 173, 2014). *Journal of Automated Reasoning*, 54:99–99, 01 2015. doi:10.1007/s10817-014-9315-z.
- [14] Matt Kaufmann, J. Strother Moore, and Panagiotis Manolios. *Computer-Aided Reasoning: An Approach*. Kluwer Academic Publishers, USA, 2000.
- [15] Gerwin Klein, June Andronick, Kevin Elphinstone, Toby Murray, Thomas Sewell, Rafal Kolanski, and Gernot Heiser. Comprehensive formal verification of an os microkernel. *ACM Trans. Comput. Syst.*, 32(1), feb 2014. doi:10.1145/2560537.
- [16] K. Rustan M. Leino. Automating induction with an smt solver. In *International Conference on Verification, Model Checking and Abstract Interpretation*, 2012. URL: <https://api.semanticscholar.org/CorpusID:6533080>.
- [17] K. Rustan M. Leino and Clément Pit-Claudel. Trigger selection strategies to stabilize program verifiers. In Swarat Chaudhuri and Azadeh Farzan, editors, *Computer Aided Verification: 28th International Conference, CAV 2016, Toronto, ON, Canada, July 17-23, 2016, Proceedings, Part I*, volume 9779 of *Lecture Notes in Computer Science*, pages 361–381. Springer International Publishing, July 2016. URL: <https://pit-claudel.fr/clement/papers/dafny-trigger-selection-CAV16.pdf>, doi:10.1007/978-3-319-41528-4\_20.
- [18] Guido Martínez, Danel Ahman, Victor Dumitrescu, Nick Giannarakis, Chris Hawblitzel, Cătălin Hrițcu, Monal Narasimhamurthy, Zoe Paraskevopoulou, Clément Pit-Claudel, Jonathan Protzenko, Tahina Ramananandro, Aseem Rastogi, and Nikhil Swamy. Meta-fstar: Proof automation with smt, tactics, and metaprograms. In Luis Caires, editor, *Programming Languages and Systems*, pages 30–59, Cham, 2019. Springer International Publishing.
- [19] Guillaume Melquiond and Raphaël Rieu-Helft. A Why3 Framework for Reflection Proofs and its Application to GMP’s Algorithms. In Didier Galmiche, Stephan Schulz, and Roberto Sebastiani, editors, *9th International Joint Conference on Automated Reasoning*, number 10900 in *Lecture Notes in Computer Science*, pages 178–193, Oxford, United Kingdom, July 2018. URL: <https://inria.hal.science/hal-01699754>, doi:10.1007/978-3-319-94205-6\_13.
- [20] Dale Miller. Robert s. boyer and j strother moore. a computational logic. acm monograph series. academic press, new york etc. 1979, xiv 397 pp. - robert s. boyer and j strother moore. a computational logic handbook. perspectives in computing, vol. 23. academic press, boston etc. 1988, xvi 408 pp. *The Journal of Symbolic Logic*, 55(3):1302–1304, 1990. doi:10.2307/2274490.
- [21] Jean-François Monin. Proof Trick: Small Inversions. In Yves Bertot, editor, *Second Coq Workshop*, Edinburgh, United Kingdom, July 2010. Yves Bertot. URL: <https://hal.inria.fr/inria-00489412>.
- [22] Jean-François Monin and Xiaomu Shi. Handcrafted inversions made operational on operational semantics. In Sandrine Blazy, Christine Paulin-Mohring, and David Pichardie, editors, *Interactive Theorem Proving*, pages 338–353, Berlin, Heidelberg, 2013. Springer Berlin Heidelberg.
- [23] Lawrence Paulson and Jasmin Blanchette. Three years of experience with sledghammer, a practical link between automatic and interactive theorem provers. 02 2015. doi:10.29007/tnfd.
- [24] Talia Ringer, RanDair Porter, Nathaniel Yazdani, John Leo, and Dan Grossman. Proof repair across type equivalences. In *Proceedings of the 42nd ACM SIGPLAN International Conference on Programming Language Design and Implementation*, PLDI 2021, page 112–127, New York, NY, USA, 2021. Association for Computing Machinery. doi:10.1145/3453483.3454033.
- [25] Talia Ringer, Nathaniel Yazdani, John Leo, and Dan Grossman. Adapting proof automation to adapt proofs. In *Proceedings of the 7th ACM SIGPLAN International Conference on Certified Programs and Proofs*, CPP 2018, page 115–129, New York, NY, USA, 2018. Association for Computing Machinery. doi:10.1145/3167094.
- [26] William Sonnex, Sophia Drossopoulou, and Susan Eisenbach. Zeno : A tool for the automatic verification of algebraic properties of functional programs. 2010. URL: <https://api.semanticscholar.org/CorpusID:2136883>.
- [27] Matthieu Sozeau and Nicolas Oury. First-class type classes. In Otmane Ait Mohamed, César Muñoz, and Sofiène Tahar, editors, *Theorem Proving in Higher Order Logics*, pages 278–293, Berlin, Heidelberg, 2008. Springer Berlin Heidelberg.
- [28] Doug Woos, James R. Wilcox, Steve Anton, Zachary Tatlock, Michael D. Ernst, and Thomas Anderson. Planning for change in a formal verification of the raft consensus protocol. In *Proceedings of the 5th ACM SIGPLAN Conference on Certified Programs and Proofs*, CPP 2016, page 154–165, New York, NY, USA, 2016. Association for Computing Machinery. doi:10.1145/2854065.2854081.
- [29] Łukasz Czajka and C. Kaliszyk. Hammer for coq: Automation for dependent type theory. *Journal of Automated Reasoning*, 61:423 – 453, 2018.